<commit_before>{ version
, sha256
, cargoSha256
}:

{ lib
, fetchFromGitHub
, rustPlatform

, pkgconfig
, openssl
, systemd
, cmake
, perl
}:

rustPlatform.buildRustPackage rec {
  pname = "parity";
  inherit version;
  inherit cargoSha256;

  src = fetchFromGitHub {
    owner = "paritytech";
    repo = "parity-ethereum";
    rev = "v${version}";
    inherit sha256;
  };

  buildInputs = [
    pkgconfig cmake perl
    systemd.lib systemd.dev openssl openssl.dev
  ];

  cargoBuildFlags = [ "--features final" ];

  # test result: FAILED. 80 passed; 12 failed; 0 ignored; 0 measured; 0 filtered out
  doCheck = false;

  meta = with lib; {
    description = "Fast, light, robust Ethereum implementation";
    homepage = "http://parity.io";
    license = licenses.gpl3;
    maintainers = with maintainers; [ akru xrelkd ];
    platforms = platforms.linux;
  };
}
<commit_msg>parity: Update buildInputs and nativeBuildInputs<commit_after>{ version
, sha256
, cargoSha256
}:

{ lib
, fetchFromGitHub
, rustPlatform

, cmake
, openssl
, pkgconfig
, systemd
}:

rustPlatform.buildRustPackage rec {
  pname = "parity";
  inherit version;
  inherit cargoSha256;

  src = fetchFromGitHub {
    owner = "paritytech";
    repo = "parity-ethereum";
    rev = "v${version}";
    inherit sha256;
  };

  nativeBuildInputs = [ cmake pkgconfig ];

  buildInputs = [ openssl systemd ];

  cargoBuildFlags = [ "--features final" ];

  # test result: FAILED. 80 passed; 12 failed; 0 ignored; 0 measured; 0 filtered out
  doCheck = false;

  meta = with lib; {
    description = "Fast, light, robust Ethereum implementation";
    homepage = "http://parity.io";
    license = licenses.gpl3;
    maintainers = with maintainers; [ akru xrelkd ];
    platforms = platforms.linux;
  };
}

{'fraction_non_alphanumeric': 0.11601584606677985, 'fraction_numerical': 0.022637238256932653, 'mean_word_length': 3.376237623762376, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1880221', 'n_tokens_mistral': 631, 'n_tokens_neox': 586, 'n_words': 155}